

6.11 Example K: User Defined Inelastic Material Model

Sample Input File For A User Defined Material Model

Problem Summary:

Load Type:	Thermomechanical
Load History:	Cyclic
Load Control:	Strain
Load History Data:	Cool-down from 400°C to 23°C , hold temperature during mechanical loading $\dot{T} = 0.419 \text{ } ^\circ\text{C/sec}$ $\dot{\epsilon} = 1.667 \times 10^{-4} / \text{sec}$, $\epsilon_{max} = 0.015$, $\epsilon_{min} = 0$. $\Delta t_{thermal} = 0.5 \text{ sec.}; \Delta t_{mech} = 0.05 \text{ sec.}$
Micromechanics model:	Double Periodicity
Fiber Packing Arrangement:	User Input 4x2 RVE with 2 fibers
Integration Algorithm:	Forward Euler
Constituent Material Model:	Fiber 1: User Model (elastic) Fiber 2: User Model (elastic) Matrix: User Model (Bodner-Partom via USRMAT)
Constituents:	Fiber 1: Imaginary material, user function material properties. Elastic modulus is a function of longitudinal strain: $E_{new} = 700 - 80(\epsilon_{11})(E_{previous})$ with units of MPa
	Fiber 2: Imaginary material, temperature dependent properties input manually.
	Matrix: Imaginary material, user function temperature dependent material properties. (see USRFUN subroutine)

☞ **Note:** The purpose here is to demonstrate how to use the various user definable subroutines, i.e., **USRMAT**, **USRFUN**, **USRFORMDE**, **USRCPEVAL** see section **4.2.12** for more information.

test of user subroutines/input
*PRINT
 NPL=10 %
*LOAD
 LCON=3 LOP=1 LSS=1 %
*MECH
 NPTW=3 TI=0.,900.,990. LO=0.,0.,0.015 %
*THERM
 NPTT=3 TI=0.,900.,990. TE=400.,23.,23. %
*MODEL
 MOD=1 %
*SOLVER
 NTF=1 NPTS=3 TIM=0.,900.,990 STP=0.5,0.05 %
*FIBER
 NFIBS=2
 NF=1 MF=99 NDPT=2 MAT=U IFM=2
 NF=2 MF=99 NDPT=2 MAT=U IFM=1 NPE=2 NPV=0
 NTP=3
 TEM=21.,200.,400.
 E1=314.1E9,293.2E9,253.0E9
 E2=0.33,0.33,0.33
 ALPA=4.5E-6,5.3E-6,6.1E-6
 ALPT=4.5E-6,5.3E-6,6.1E-6
*MATRIX
 NMATX=1
 NM=1 MM=99 NDPT=2 MAT=U IFM=2 %
*MRVE
 IDP=99
 NB=4 NG=2
 H=1.,1.,1.,1.
 L=1.,1.
 CM=M1,M1
 CM=F2,M1
 CM=M1,M1
 CM=F1,M1
*CURVE
 NP=2 %
*MACRO
 NT=1
 NC=1 X=1 Y=7 NAM=apdx-k
*END




Figure was obtained from the x-y plot data file produced by the present example.

The USRMAT subroutine

The USRMAT subroutine is used here to implement the Bodner-Partom Viscoplastic model currently available in **MAC/GMC**

```

c#####
c      SUBROUTINE USRMAT(DSA, SA, PE, PV, D, LOCTISO, TIME, TSTEP,
c      &          CTEMP, DTEMPR, NIO, NE, NV, NS, MN, CDUM, DMGF,
c      &          NEP, NVP, NSASIZE)

c      purpose: user material constitutive model for determination of
c              the inelastic strain and state variable rates
c              (used when ncnd = 99)

c      IMPLICIT DOUBLE PRECISION (A - H, O - Z)

CHARACTER*2 CDUM
DIMENSION SS(6), S(6), R(6)
DIMENSION DSA(NSASIZE), SA(NSASIZE)
DIMENSION PV(NVP), PE(NEP), D(3)

*****
c  note: 1) in this subroutine, [SA] and [DSA] contain the
c         micro (subcell) quantities for aboudi's micromechanics model
c
c         2) arrangement of [dsa] & [sa] arrays:
c             variable           location
c
c             +-----+
c             | strain rate       (1-6)  (contains ENGINEERING shears)
c             +-----+
c             | stress rate        (7-12)
c             +-----+
c             | inelastic
c             | strain rate        (13-18) (contains ENGINEERING shears)
c             +-----+
c             | 12 "slots"         (19-30)
c             | for state variables
c             +-----+
c             | thermal strain rate (31-36)
c             +-----+
*****
c  on entry:
c      SA      - vector of total (integrated) quantities (see above)
c      PE(NE)  - vector of elastic constants for material MN
c                  (where NE = # of elastic constants --> 9 MAX)
c      PV(NV)  - vector of viscoplastic constants for material MN
c                  (where NV = # of viscoplastic constants --> 19 MAX)
c      D(3)    - vector of direction cosines (for models 3, 7, & 9)
c      LOCTISO - flag indicating if ANY material exhibits local
c                  transverse isotropy (and global anisotropy)
c                  = 0 - all materials are at most globally transversely
c                  isotropic (D not used)

```

```

c           = 1 - at least one material is locally transversely
c           isotropic (D used)
c   TIME      - current time
c   TSTEP     - current time step
c   CTEMP     - current temperature
c   DTEMPR    - time rate of change of temperature
c   NIO       - unit number of .out file
c   NE        - # of elastic constants --> 9 MAX
c   NV        - # of viscoplastic constants --> 19 MAX
c   NS        - subcell number
c   MN        - material number
c   CDUM      - material character/number designation
c                  (i.e. F1 = fiber #1)
c   DMGF      - damage factor - if damage is included the user
c                  should multiply material stiffness terms by DMGF
c                  when using such terms in his inelastic model.
c
c   expected on exit:
c   DSA       - vector of rate quantities (see above)
*****

```

```

IF( NE .GT. NEP ) THEN
  CALL FATALERROR(NIO)
  WRITE(NIO, *) 'TO MANY ELASTIC PROPERTIES FOR MATERIAL # ', MN
  WRITE(NIO, *) ' # USED      = ', NE
  WRITE(NIO, *) ' # ALLOCATED = ', NEP
  STOP
ENDIF

IF( NV .GT. NVP ) THEN
  CALL FATALERROR(NIO)
  WRITE(NIO, *) 'TO MANY NON-LINEAR PROPERTIES FOR MATERIAL # ',
&             MN
  WRITE(NIO, *) ' # USED      = ', NV
  WRITE(NIO, *) ' # ALLOCATED = ', NVP
  STOP
ENDIF

*****
*          BEGIN USER EDITS
*****
C   WRITE(6, *) ' PV=', PV
C   WRITE(6, *) ' PE=', PE

C-----
c   MATERIALS #1 & #2 ARE ELASTIC
c   (set inelastic strain rates to zero)
C-----

IF ((MN .EQ. 1) .OR. (MN .EQ. 2)) THEN
  DSA(13) = 0
  DSA(14) = 0
  DSA(15) = 0

```

```

        DSA(16) = 0
        DSA(17) = 0
        DSA(18) = 0
        RETURN
    ENDIF

C-----
C      MATERIAL #3 --> USE BODNER-PARTOM
C-----

C-----
C      copy appropriate viscoplastic material constants
C-----

        IF( NV .LT. 6 ) THEN
            CALL FATALERROR(NIO)
            WRITE(NIO, *) ' NOT ENOUGH PV SPACE: NV =', NV
            STOP
        ENDIF
        D0 = PV(1)
        Z0 = PV(2)
        Z1 = PV(3)
        BM = PV(4)
        AN = PV(5)
        Q = PV(6)

C-----
C      copy stress from [sa] to [ss]
C-----

        SS(1) = SA(7)
        SS(2) = SA(8)
        SS(3) = SA(9)
        SS(4) = SA(10)
        SS(5) = SA(11)
        SS(6) = SA(12)

C-----
C      compute the deviatoric stress [s] in the subcell
C-----

        TEMP = (SS(1) + SS(2) + SS(3)) / 3.0
        S(1) = SS(1) - TEMP
        S(2) = SS(2) - TEMP
        S(3) = SS(3) - TEMP
        S(4) = SS(4)
        S(5) = SS(5)
        S(6) = SS(6)

C-----
C-----
```

$$AJ2 = 0.5 * (S(1)^2 + S(2)^2 + S(3)^2) + S(4)^2 + S(5)^2 + S(6)^2$$

$$SQ3AJ = DSQRT(SS(1)^2 + SS(2)^2 + SS(3)^2 + 2 * (SS(4)^2 + SS(5)^2 + SS(6)^2))$$

$$SQ2 = 1.414215$$

$$\text{IF } (SQ3AJ .EQ. 0.0) \text{ THEN}$$

$$\quad \text{CALL ZEROR(R, 6)}$$

$$\text{ELSE}$$

```

        R(1) = SS(1) / SQ3AJ
        R(2) = SS(2) / SQ3AJ
        R(3) = SS(3) / SQ3AJ
        R(4) = SQ2 * SS(4) / SQ3AJ
        R(5) = SQ2 * SS(5) / SQ3AJ
        R(6) = SQ2 * SS(6) / SQ3AJ
    ENDIF
C-----
c  if d0=0 then assume elastic and zero-out
c  [dsa(13-30)] (inelastic strain rate and
c  internal variable rates), then return
C-----
        IF (D0 .EQ. 0) THEN
            DO 100 JJ = 13, 30
                DSA(JJ) = 0.0
100         CONTINUE
            RETURN
C-----
c  inelastic
C-----
        ELSE
            ZEF = Z0 + Q * SA(20) + (1 - Q) * (R(1) * SA(21) + R(2) *
&             SA(22) + R(3) * SA(23) + R(4) * SA(24) + R(5)
&             * SA(25) + R(6) * SA(26))

            IF (AJ2 .EQ. 0.0) THEN
                AL = 0.0
            ELSE
                ARG1 = ZEF**2.0 / (3.0 * AJ2)

                IF (ARG1 .GT. 1E6) ARG1 = 1E6
                CON = .5 * (AN + 1.0) / AN
                ARG = CON * (ARG1)**AN

                IF (ARG .GT. 50.0) ARG = 50.0
                AL = D0 / (DEXP(ARG) * DSQRT(AJ2))
            ENDIF
C-----
c  inelastic strain rates
C-----
            DSA(13) = AL * S(1)
            DSA(14) = AL * S(2)
            DSA(15) = AL * S(3)
            DSA(16) = 2 * AL * S(4)
            DSA(17) = 2 * AL * S(5)
            DSA(18) = 2 * AL * S(6)

C-----
c  plastic work rate
C-----
            WPD = S(1) * DSA(13) + S(2) * DSA(14) + S(3) * DSA(15) + S(4) *
&             DSA(16) + S(5) * DSA(17) + S(6) * DSA(18)
C-----
c  state variable rates

```

```

C-----
      DSA(19) = WPD
      Z0M = BM / Z0
      ZD = Z0M * (Z1 - ZEF) * WPD
      DSA(20) = ZD

      DSA(21) = ZD * R(1)
      DSA(22) = ZD * R(2)
      DSA(23) = ZD * R(3)
      DSA(24) = ZD * R(4)
      DSA(25) = ZD * R(5)
      DSA(26) = ZD * R(6)
ENDIF

*****
*          END USER EDITS
*****
RETURN
END

```

The USRFUN subroutine

The USRFUN subroutine is used here to allow material properties to be entered in functional form instead of having to be linearly interpolated within **MAC/GMC**

```

C#####
SUBROUTINE USRFUN(MN, TIME, TSTEP, CTEMP, DTEMPR, SA, DSA,
&      DOLD, PEM, PVM, D, LOCTISO, ALPA, ALPT,
&      NE, NV, NMTS, NEP, NVP, NSASIZE)

C      purpose: user subroutine to allow elastic and viscoplastic
C      material properties to be functions of TEMP or
C      field variables. Used for user defined functional
C      form material properties, that is, when:
C      (mat .eq. 'U') .and. (ifm .eq. 2)
C
C      note: can be used in conjunction with a provided material
C      constitutive model, or a constitutive model input
C      by the user in USRMAT

IMPLICIT DOUBLE PRECISION (A - H, O - Z)

DIMENSION DOLD(6, 6), D(3)

DIMENSION PEM(NEP, NMTS), PVM(NVP, NMTS)
DIMENSION ALPA(NMTS), ALPT(NMTS)
DIMENSION DSA(NSASIZE), SA(NSASIZE)

```

```
*****
c note: 1) in this subroutine, [SA] and [DSA] contain the
c         micro (subcell) quantities for aboudi's micromechanics model
c
c         2) arrangement of [dsa] & [sa] arrays:
c             variable           location
c
c             +-----+
c             | strain rate          (1-6)  (contains ENGINEERING shears)
c             +-----+
c             | stress rate          (7-12)
c             +-----+
c             | inelastic
c             | strain rate          (13-18) (contains ENGINEERING shears)
c             +-----+
c             | 12 "slots"           (19-30)
c             | for state variables
c             +-----+
c             | thermal strain rate (31-36)
c             +-----+
*****
c NOTE: quantities in [SA] and [DSA] are SUBCELL quantities - the
c values on entry are for the first subcell containing material
c # MN - the values on exit of this subroutine will be applied to
c ALL SUBCELLS containing material # MN. It is thus recommended
c that, if using the field variables, you assign the appropriate
c material # to ONE SUBCELL ONLY. Use of [SA] and [DSA] in this
c context in conjunction with bending in laminate theory will
c result in erroneous results as field variables become dependent
c on through-thickness position (while the material # does not).
*****
c on entry:
c     MN      - material number
c     TIME    - current time
c     TSTEP   - current time step
c     CTEMP   - current temperature
c     DTEMPR  - time rate of change of temperature
c     SA      - vector of total (integrated) quantities (see above)
c     DSA     - vector of rate quantities (see above)
c     DOLD(6, 6) - previous elastic material stiffness matrix
c     PEM(NE, MN) - vector of previous elastic constants for material
c                   # MN (where NE = # of elastic constants --> 9 MAX)
c     PVM(NV, MN) - vector of previous viscoplastic constants for
c                   material # MN
c                   (where NV = # of viscoplastic constants --> 19 MAX)
c
c expected on exit:
c     PEM(NE, MN) - vector of current elastic constants for material MN
c     PVM(NV, MN) - vector of current viscoplastic constants for
c                   material MN
c     D(3)       - vector of direction cosines
c                   (required for models 3, 7, & 9)
c     LOCTISO   - flag indicating if ANY material exhibits local
c                   transverse isotropy (and global anisotropy)
c                   = 0 - all materials are at most globally transversely
*****
```

```

C           isotropic (D not used)
C           = 1 - at least one material is locally transversely
C                   isotropic (D used)
C   ALPA(MN)    - longitudinal cte for material MN
C   ALPT(MN)    - transverse cte for material MN
C   NE          - NUMBER OF ELASTIC PROPERTIES USED
C   NV          - NUMBER OF VISCOPLASTIC PROPERTIES USED
C
*****  

*****  

*               BEGIN USER EDITS  

*****  

C   WRITE(6, *) 'TOP OF USRFUN'  

C-----  

C MATERIAL # 1: E = FUNCTION OF STRAIN & PREVIOUS E  

C                 FOR USE WITH USER CONSTITUTIVE MODEL  

C-----  

IF (MN .EQ. 1) THEN  

  IF (SA(1) .GT. 0) THEN  

    EAOLD = PEM(1, MN)  

    EA = 700.E9 - EAOLD * SA(1) * 80.0  

  ELSE  

    EA = 700.E9  

  ENDIF  

  ET = EA  

  FNA = 0.41  

  FNT = 0.41  

  GA = EA / (2.0 * (1.0 + FNA))  

  ALPA(MN) = 4.5E-6  

  ALPT(MN) = 4.5E-6  

  NE = 5  

  PEM(1, MN) = EA  

  PEM(2, MN) = ET  

  PEM(3, MN) = FNA  

  PEM(4, MN) = FNT  

  PEM(5, MN) = GA  

C-----  

C MATERIAL # 2: PROPERTIES = INTERPOLATED AT INPUT TEMPERATURES  

C                 THIS SUBROUTINE IS NOT CALLED FOR MN #2 SINCE THE  

C                 MATERIAL PROPERTIES ARE NOT FUNCTIONAL FORM  

C-----  

C-----  

C MATERIAL # 3: PROPERTIES = LINEAR FUNCTION OF TEMPERATURE  

C                 FOR USE WITH USER CONSTITUTIVE MODEL  

C-----  


```

```
ELSEIF (MN .EQ. 3) THEN

    IF (CTEMP .LT. 21.0) CTEMP = 21.0
    IF (CTEMP .GT. 400.) CTEMP = 400.

C -- ELASTIC
    E = 72.4E9 - 81.53E6 * (CTEMP - 21.)
    FN = 0.33 + 7.916E-5 * (CTEMP - 21.)
    ALP = 22.5E-6 + 3.958E-9 * (CTEMP - 21.)

    NE = 5
    PEM(1, MN) = E
    PEM(2, MN) = E
    PEM(3, MN) = FN
    PEM(4, MN) = FN
    PEM(5, MN) = E / (2. * (1. + FN))
    ALPA(MN) = ALP
    ALPT(MN) = ALP

C -- VISCOPLASTIC
    D0 = 1.E4
    Z0 = 340.E6
    Z1 = 435.E6
    BM = 300.0
    AN = 10.0 - 0.02493 * (CTEMP - 21.)
    Q = 1.0

    NV = 6
    PVM(1, MN) = D0
    PVM(2, MN) = Z0
    PVM(3, MN) = Z1
    PVM(4, MN) = BM
    PVM(5, MN) = AN
    PVM(6, MN) = Q

ENDIF

*****
*          END USER EDITS
*****
RETURN
END
```

The USRFORMDE subroutine

The USRFORMDE subroutine is used here to form the elastic material stiffness matrix in **MAC/GMC** when a user defined constitutive model is used.

```

c#####
c      SUBROUTINE USRFORMDE(MN, PEM, PVM, D, LOCTISO, DNEW,
c      &           NE, NV, NEP, NVP, NMITS)
c
c      purpose: user subroutine to allow formation of material stiffness
c      matrices based on a user constitutive model, which may use
c      arbitrary material properties (used when ncnd = 99)
c
c      IMPLICIT DOUBLE PRECISION (A - H, O - Z)
c
c      DIMENSION DNEW(6, 6)
c      DIMENSION PEM(NEP, NMITS), PVM(NVP, NMITS)
c      DIMENSION D(3)
c
*****  

c      on entry:  

c      MN          - material number  

c      PEM(NE, MN) - vector of elastic constants for material  

c                      # MN (where NE = # of elastic constants --> 9 MAX)  

c      PVM(NV, MN) - vector of viscoplastic constants for material # MN  

c                      (where NV = # of viscoplastic constants --> 19 MAX)  

c      D(3)         - vector of direction cosines  

c                      (required for models 3, 7, & 9)  

c      LOCTISO     - flag indicating if ANY material exhibits local  

c                      transverse isotropy (and global anisotropy)  

c                      = 0 - all materials are at most globally transversely  

c                            isotropic (D not used)  

c                      = 1 - at least one material is locally transversely  

c                            isotropic (D used)  

c
c      expected on exit:  

c      DNEW(6, 6) - current elastic material stiffness matrix
*****  

*           BEGIN USER EDITS
*****  

C      WRITE(6, *) 'TOP OF USRFORMDE'  

C NOTE: In the examples shown here, standard engineering material
C       elastic constants (E, nu) are used. However, the user is
C       free to employ any material elastic constants with his
C       constitutive model (i.e., bulk modulus, etc.). Thus, the
C       user must provide the equations to determine the stiffness
C       components required by GMC from his elastic constants.  

C-----  

C      MATERIAL # 1: E = FUNCTION OF STRAIN & PREVIOUS E

```

```

C           FOR USE WITH USER CONSTITUTIVE MODEL
C-----
IF (MN .EQ. 1) THEN

    EA = PEM(1, MN)
    ET = PEM(2, MN)
    FNA = PEM(3, MN)
    FNT = PEM(4, MN)
    GA = PEM(5, MN)

    DO 100 I = 1, 6
        DO 100 J = 1, 6
            DNEW(I, J) = 0
100    CONTINUE

    GT = 0.5 * ET / (1 + FNT)
    FK = 0.25 * EA / (0.5 * (1 - FNT) * (EA / ET) - FNA**2)

    DNEW(1, 1) = EA + 4.0 * FK * FNA**2
    DNEW(2, 1) = 2.0 * FK * FNA
    DNEW(3, 1) = 2.0 * FK * FNA

    DNEW(1, 2) = 2.0 * FK * FNA
    DNEW(2, 2) = FK + GT
    DNEW(3, 2) = FK - GT

    DNEW(1, 3) = 2.0 * FK * FNA
    DNEW(2, 3) = FK - GT
    DNEW(3, 3) = FK + GT

    DNEW(4, 4) = GT
    DNEW(5, 5) = GA
    DNEW(6, 6) = GA

C-----
C MATERIAL # 2: PROPERTIES = INTERPOLATED AT TEMPERATURES
C                   FOR USE WITH USER CONSTITUTIVE MODEL
C-----
ELSEIF (MN .EQ. 2) THEN

    E = PEM(1, MN)
    FN = PEM(2, MN)
    GA = E / (2. * (1. + FN))

    DO 320 I = 1, 6
        DO 320 J = 1, 6
            DNEW(I, J) = 0
320    CONTINUE

    GT = 0.5 * E / (1 + FN)
    FK = 0.25 * E / (0.5 * (1 - FN) - FN**2)

    DNEW(1, 1) = E + 4.0 * FK * FN**2
    DNEW(2, 1) = 2.0 * FK * FN

```

```
DNEW(3, 1) = 2.0 * FK * FN
DNEW(1, 2) = 2.0 * FK * FN
DNEW(2, 2) = FK + GT
DNEW(3, 2) = FK - GT

DNEW(1, 3) = 2.0 * FK * FN
DNEW(2, 3) = FK - GT
DNEW(3, 3) = FK + GT

DNEW(4, 4) = GT
DNEW(5, 5) = GA
DNEW(6, 6) = GA

C-----
C MATERIAL # 3: PROPERTIES = LINEAR FUNCTION OF TEMPERATURE
C                   FOR USE WITH USER CONSTITUTIVE MODEL
C-----

ELSEIF (MN .EQ. 3) THEN

    E = PEM(1, MN)
    FN = PEM(3, MN)
    GA = PEM(5, MN)

    DO 300 I = 1, 6
        DO 300 J = 1, 6
            DNEW(I, J) = 0
300    CONTINUE

    GT = 0.5 * E / (1 + FN)
    FK = 0.25 * E / (0.5 * (1 - FN) - FN**2)

    DNEW(1, 1) = E + 4.0 * FK * FN**2
    DNEW(2, 1) = 2.0 * FK * FN
    DNEW(3, 1) = 2.0 * FK * FN

    DNEW(1, 2) = 2.0 * FK * FN
    DNEW(2, 2) = FK + GT
    DNEW(3, 2) = FK - GT

    DNEW(1, 3) = 2.0 * FK * FN
    DNEW(2, 3) = FK - GT
    DNEW(3, 3) = FK + GT

    DNEW(4, 4) = GT
    DNEW(5, 5) = GA
    DNEW(6, 6) = GA

ENDIF
*****
*          END USER EDITS
*****
RETURN
END
```

The USRCPEVAL subroutine

The USRCPEVAL subroutine is used here to form the time derivative of the material stiffness matrix in **MAC/GMC**

```

c#####
      SUBROUTINE USRCPEVAL(DSA, SA, MN, TIME, TSTEP, CTEMP, DTEMPR,
     &           DNEW, DOLD, PEM, PVM, D, LOCTISO, ALPA, ALPT, DDOT,
     &           NE, NV, NMTS, NEP, NVP, NSASIZE)

c      purpose: user subroutine to allow formation of the TIME
c              derivative of the material stiffness matrix.
c              this subroutine is used when:
c                  a) material properties are user defined and functional
c                      form. That is: (mat .eq. 'U') .and. (ifm .eq. 2)
c                  b) the constitutive model is user-defined, and the
c                      material properties are not functional form, and
c                      the material properties are temperature-dependent
c                      that is: (ncmd .eq. 99) .and. (ifm .ne. 2) .and.
c                      (ndpt .eq. 2)

c      IMPLICIT DOUBLE PRECISION (A - H, O - Z)

DIMENSION DNEW(6, 6), DOLD(6, 6)
DIMENSION DDOT(6, 6)

DIMENSION PEM(NEP, NMTS), PVM(NVP, NMTS)
DIMENSION ALPA(NMTS), ALPT(NMTS)
DIMENSION DSA(NSASIZE), SA(NSASIZE)

*****
c  note: 1) in this subroutine, [SA] and [DSA] contain the
c          micro (subcell) quantities for aboudi's micromechanics model
c
c          2) arrangement of [dsa] & [sa] arrays:
c              variable          location
c              +-----+
c              | strain rate        (1-6)  (contains ENGINEERING shears)
c              |-----+
c              | stress rate         (7-12)
c              |-----+
c              | inelastic
c              | strain rate        (13-18) (contains ENGINEERING shears)
c              |-----+
c              | 12 "slots"          (19-30)
c              | for state variables
c              |-----+
c              | thermal strain rate (31-36)
c              +-----+
***** 
c  on entry:
c      SA           - vector of total (integrated) quantities (see above)

```

```

C      DSA          - vector of rate quantities (see above)
C      MN           - material number
C      TIME         - current time
C      TSTEP         - current time step
C      CTEMP         - current temperature
C      DTEMPR        - time rate of change of temperature
C      DNEW(6, 6)    - current elastic material stiffness matrix
C      DOLD(6, 6)    - previous elastic material stiffness matrix
C      PEM(NE, MN)   - vector of current elastic constants for material MN
C                      (where NE = # of elastic constants --> 9 MAX)
C      PVM(NV, MN)   - vector of current viscoplastic constants for
C                      material MN
C                      (where NV = # of elastic constants --> 19 MAX)
C      D(3)          - vector of direction cosines
C                      (required for models 3, 7, & 9) local
C      LOCTISO       - flag indicating if ANY material exhibits
C                      transverse isotropy (and global anisotropy)
C                      = 0 - all materials are at most globally transversely
C                      isotropic (D not used)
C                      = 1 - at least one material is locally transversely
C                      isotropic (D used)
C      ALPA(MN)      - longitudinal cte for material MN
C      ALPT(MN)      - transverse cte for material MN
C
C      expected on exit:
C      DDOT(6, 6)    - derivative with respect to TIME of stiffness matrix
*****

```

```

*****
*                               BEGIN USER EDITS
*****

```

```

DIMENSION DHOLD(6, 6, 3), DNEW1(6, 6), PEM1(9, 10), PVM1(19, 10)
DIMENSION ALPA1(10), ALPT1(10)

```

```
C      WRITE(6, *) 'TOP OF USRCPEVAL'
```

```

C-----
C MATERIAL # 1: E = FUNCTION OF STRAIN & PREVIOUS E
C                  FOR USE WITH USER CONSTITUTIVE MODEL
C-----

```

```

IF (MN .EQ. 1) THEN

C --- method one - simply divide change in stiffness components by TSTEP
DO 200 I = 1, 6
    DO 200 J = 1, 6
        DDOT(I, J) = (DNEW(I, J) - DOLD(I, J)) / TSTEP
200     CONTINUE

```

```

C-----
C MATERIAL # 2: PROPERTIES = INTERPOLATED AT TEMPERATURES
C                  FOR USE WITH USER CONSTITUTIVE MODEL
C-----
```

```

C-----
      ELSEIF (MN .EQ. 2) THEN

c --- method 2 - if material props are only function of temperature,
c --- calculate stiffness at CTEMP - 0.5 & CTEMP + 0.5, difference
c --- equals change per temp, multiply by DTEMPR equals change per time
      IF (CTEMP .LT. 21.0) CTEMP = 21.0
      IF (CTEMP .GT. 400.) CTEMP = 400.

      DO 300 I = 1, 6
          DO 300 J = 1, 6
              DO 300 K = 1, 2
                  DHOLD(I, J, K) = 0
300      CONTINUE

      DO 420 K = 1, 2

          IF (K .EQ. 1) CTEMP1 = CTEMP - 0.5
          IF (K .EQ. 2) CTEMP1 = CTEMP + 0.5

c --- this subroutine interpolates properties for material MN at
c --- the temperature CTEMP1
          CALL INTTMR(MN, CTEMP1, PEM1, ALPA1, ALPT1, PVM1)

c --- calculate the stiffness components
          CALL USRFORMDE(MN, PEM1, PVM, D, LOCTISO, DNEW1,
                         & NE, NV, NEP, NVP, NMRS)
c --- note: for internal constitutive models (ncmd .ne. 99), use
c --- subroutine FORMDE to calculate stiffness components.
c --- syntax: CALL FORMDE(MN, PEM1, D)

          DO 480 I = 1, 6
              DO 480 J = 1, 6
                  DHOLD (I, J, K) = DNEW1(I, J)
480      CONTINUE
420      CONTINUE

          DO 520 I = 1, 6
              DO 520 J = 1, 6
                  DDOT(I, J) = (DHOLD(I, J, 2) - DHOLD(I, J, 1)) * DTEMPR
520      CONTINUE

C-----
C MATERIAL # 3: PROPERTIES = LINEAR FUNCTION OF TEMPERATURE
C                 FOR USE WITH USER CONSTITUTIVE MODEL
C-----

      ELSEIF (MN .EQ. 3) THEN

c --- method 2 - if material props are only function of temperature,
c --- calculate stiffness at CTEMP - 0.5 & CTEMP + 0.5, difference
c --- equals change per temp, multiply by DTEMPR equals change per time
      IF (CTEMP .LT. 21.0) CTEMP = 21.0
      IF (CTEMP .GT. 400.) CTEMP = 400.

```

```
DO 330 I = 1, 6
    DO 330 J = 1, 6
        DO 330 K = 1, 2
            DHOLD(I, J, K) = 0
330      CONTINUE

DO 400 K = 1, 2

    IF (K .EQ. 1) CTEMP1 = CTEMP - 0.5
    IF (K .EQ. 2) CTEMP1 = CTEMP + 0.5

c --- determine material props at the appropriate temperature
    CALL USRFUN(MN, TIME, TSTEP, CTEMP1, DTEMPR, SA, DSA,
&             DOLD, PEM1, PVM1, D, LOCTISO, ALPA1, ALPT1,
&             NE, NV, NMITS, NEP, NVP, NSASIZE)

c --- calculate the stiffness components
    CALL USRFORMDE(MN, PEM1, PVM, D, LOCTISO, DNEW1,
&                 NE, NV, NEP, NVP, NMITS)
c --- note: for internal constitutive models (ncmd .ne. 99), use
c --- subroutine FORMDE to calculate stiffness components.
c --- syntax: CALL FORMDE(MN, PEM1, D)

DO 450 I = 1, 6
    DO 450 J = 1, 6
        DHOLD (I, J, K) = DNEW1(I, J)
450      CONTINUE
400      CONTINUE

DO 500 I = 1, 6
    DO 500 J = 1, 6
        DDOT(I, J) = (DHOLD(I, J, 2) - DHOLD(I, J, 1)) * DTEMPR
500      CONTINUE

ENDIF
*****
*           END USER EDITS
*****
```

RETURN
END